Fractions continues

O.Marguin - 11 mai 2011

Exemples

La fonction convert avec l'option confrac donne le début du développement en fraction continue, ainsi que les premières réduites. Attention, c'est un procédé numérique : les dernières valeurs peuvent être erronées. Il y a des fonctions spécialisées (par exemple cfrac) dans le package numtheory, que nous n'utiliserons pas.

```
> restart:
> convert(evalf(sqrt(2)),confrac,'r');r;
```

$$[1, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 1, 1, 14]$$

$$\left[1, \frac{3}{2}, \frac{7}{5}, \frac{17}{12}, \frac{41}{29}, \frac{99}{70}, \frac{239}{169}, \frac{577}{408}, \frac{1393}{985}, \frac{3363}{2378}, \frac{8119}{5741}, \frac{19601}{13860}, \frac{27720}{19601}, \frac{47321}{33461}, \frac{690214}{488055}\right]$$

```
> convert(evalf((1+sqrt(5))/2),confrac,'r');r;
```

$$[1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 2, 5]$$

$$\left[1, 2, \frac{3}{2}, \frac{5}{3}, \frac{8}{5}, \frac{13}{8}, \frac{21}{13}, \frac{34}{21}, \frac{55}{34}, \frac{89}{55}, \frac{144}{89}, \frac{233}{144}, \frac{377}{233}, \frac{610}{377}, \frac{987}{610}, \frac{1597}{987}, \frac{2584}{1597}, \frac{4181}{2584}, \frac{6765}{4181}, \frac{10946}{6765}, \frac{17711}{10946}, \frac{46368}{28657}, \frac{249551}{154231}\right]$$

```
> convert(evalf(Pi),confrac,'r');r;
```

$$[3, 7, 15, 1, 293, 11]$$

$$\left[3, \frac{22}{7}, \frac{333}{106}, \frac{355}{113}, \frac{104348}{33215}, \frac{1148183}{365478}\right]$$

On reconnaît les approximations d'Archimède et de Metius (2ème et 4ème réduites).

```
>
```

Cas d'un nombre rationnel

```
> restart:
> rat2fc:=proc(x) # calcule le développement en fraction continue du rationnel x
      local a,b,q,r;
      a:=numer(x);b:=denom(x);q:=iquo(a,b);r:=irem(a,b);
      if r=0 then [q] else [q,op(rat2fc(b/r))] end if
  end proc:
> rat2fc(355/113);
```

$$[3, 7, 16]$$

```
> fc2rat:=proc(L) # fonction inverse
      if nops(L)=1 then L[1] else L[1]+1/fc2rat(L[2..nops(L)]) end if
  end proc:
> fc2rat([3,7,16]);
```

$$\frac{355}{113}$$

Cas d'un développement infini périodique

On cherche x de développement périodique (6,5,4,3,2,1,6,5,4,3,2,1,...).
x est solution de l'équation :

```
> equ1:=x=fc2rat([6,5,4,3,2,1,x]);
```

$$equ1 := x = 6 + \cfrac{1}{5 + \cfrac{1}{4 + \cfrac{1}{3 + \cfrac{1}{2 + \cfrac{1}{1 + \cfrac{1}{x}}}}}}$$

```
> equ2:=normal(equ1);
```

$$equ2 := x = \frac{1393\,x + 972}{225\,x + 157}$$

```
> P:=x*denom(rhs(equ2))-numer(rhs(equ2));
```

$$P := x\,(225\,x+157) - 1393\,x - 972$$

```
> P:=sort(expand(P),x);
```

$$P := 225\,x^2 - 1236\,x - 972$$

```
> s:=solve(P);
```

$$s := \frac{206}{75}+\frac{4}{75}\sqrt{4171},\ \frac{206}{75}-\frac{4}{75}\sqrt{4171}$$

x est la solution positive :

```
> s[1];
```

$$\frac{206}{75}+\frac{4}{75}\sqrt{4171}$$

```
> fc2alg:=proc(L) # automatisation du calcul
      local eq,s;
      eq:=x=fc2rat([op(L),x]);
      max(solve(eq))
  end:
> fc2alg([6,5,4,3,2,1]);
```

$$\frac{206}{75}+\frac{4}{75}\sqrt{4171}$$

Applications :

```
> fc2alg([1]);
```

$$\frac{1}{2}\sqrt{5}+\frac{1}{2}$$

Le développement en fraction continue de ϕ est donc (1,1,1,...)

```
> fc2alg([2]);
```

$$\sqrt{2}+1$$

Le développement en fraction continue de $\sqrt{2}$ est donc (1,2,2,2,...).

```
> convert(evalf(sqrt(3)),confrac);convert(evalf(sqrt(5)),confrac);convert(evalf(sqrt(
  7)),confrac);
```

$$[1, 1, 2, 1, 2, 1, 2, 1, 2, 1, 2, 1, 2, 1, 2, 1, 3, 2, 8]$$

$$[2, 4, 4, 4, 4, 4, 4, 3, 1, 21]$$

$$[2, 1, 1, 1, 4, 1, 1, 1, 4, 1, 1, 1, 4, 1, 1, 1, 4, 2]$$

```
> fc2alg([2,1]);fc2alg([4]);fc2alg([4,1,1,1]);
```

$$1+\sqrt{3}$$

$$2+\sqrt{5}$$

$$2+\sqrt{7}$$

D'où les développements respectifs de $\sqrt{3}$, $\sqrt{5}$ et $\sqrt{7}$: (1,1,2,1,2,1,2,...) , (2,4,4,4,4,...) , (2,1,1,1,4,1,1,1,4,1,1,1,4,...)

```
>
```

Cas d'un nombre quadratique

Soit $x=\frac{2}{3}+\frac{5\sqrt{13}}{7}$. On se propose de prouver que le développement en fraction continue de x est périodique à partir d'un certain rang, en se calquant sur le cours, paragraphe (1.7).

```
> restart:
> alpha:=P->floor(max(solve(P,X)));
```

$$\alpha := P \to \mathrm{floor}(\max(\mathrm{solve}(P, X)))$$

```
> phi:=P->collect(expand(-X^2*subs(X=alpha(P)+1/X,P)),X);
```

$$\phi := P \to \mathrm{collect}\left(\mathrm{expand}\left(-X^2\ \mathrm{subs}\left(X=\alpha(P)+\frac{1}{X}, P\right)\right), X\right)$$

```
> P:=a*X^2+b*X+c; phi(P);
```

$$P := a\,X^2 + b\,X + c$$

$$\left(-b\ \mathrm{floor}\left(\max\left(-\frac{b}{2\,a}+\frac{\sqrt{b^2-4\,a\,c}}{2\,a}, -\frac{b}{2\,a}-\frac{\sqrt{b^2-4\,a\,c}}{2\,a}\right)\right)\right.$$

$$- a \operatorname{floor}\left(\max\left(-\frac{b}{2a}+\frac{\sqrt{b^2-4ac}}{2a}, -\frac{b}{2a}-\frac{\sqrt{b^2-4ac}}{2a}\right)\right)^2 - c\Bigg) X^2$$
$$+\left(-b-2a \operatorname{floor}\left(\max\left(-\frac{b}{2a}+\frac{\sqrt{b^2-4ac}}{2a}, -\frac{b}{2a}-\frac{\sqrt{b^2-4ac}}{2a}\right)\right)\right) X - a$$

```
> discrim(P,X);discrim(phi(P),X);
```

$$b^2 - 4\,a\,c$$
$$b^2 - 4\,a\,c$$

Bravo Maple !
Cherchons le polynôme minimal de x :

```
> eq:=X=2/3+5/7*sqrt13;
```

$$eq := X = \frac{2}{3} + \frac{5\ \mathrm{sqrt13}}{7}$$

```
> P:=expand(solve(eq,sqrt13)^2-13);
```

$$P := \frac{49}{25} X^2 - \frac{196}{75} X - \frac{2729}{225}$$

```
> P0:=225*P;solve(P0);
```

$$P0 := 441\,X^2 - 588\,X - 2729$$
$$\frac{2}{3} + \frac{5\sqrt{13}}{7}, \frac{2}{3} - \frac{5\sqrt{13}}{7}$$

Calculons la suite des polynômes P_n du cours, en s'arrêtant dès qu'on retombe sur un polynôme déjà vu :

```
> S:=P0:Q:=phi(P0):
  while {S} union {Q}<>{S} do S:=S,Q;Q:=phi(Q) od:
  S;Q;
```

$441\,X^2 - 588\,X - 2729$, $524\,X^2 - 2058\,X - 441$, $289\,X^2 - 2134\,X - 524$, $1301\,X^2 - 1912\,X - 289$, $900\,X^2 - 690\,X - 1301$,
$1091\,X^2 - 1110\,X - 900$, $919\,X^2 - 1072\,X - 1091$, $1244\,X^2 - 766\,X - 919$, $441\,X^2 - 1722\,X - 1244$,
$1076\,X^2 - 1806\,X - 441$, $1171\,X^2 - 346\,X - 1076$, $251\,X^2 - 1996\,X - 1171$, $1075\,X^2 - 2020\,X - 251$,
$1196\,X^2 - 130\,X - 1075$, $9\,X^2 - 2262\,X - 1196$, $1949\,X^2 - 2256\,X - 9$, $316\,X^2 - 1642\,X - 1949$, $425\,X^2 - 2150\,X - 316$,
$441\,X^2 - 2100\,X - 425$, $1769\,X^2 - 1428\,X - 441$, $100\,X^2 - 2110\,X - 1769$, $1979\,X^2 - 2090\,X - 100$,
$211\,X^2 - 1868\,X - 1979$, $1700\,X^2 - 1930\,X - 211$, $441\,X^2 - 1470\,X - 1700$

$$524\,X^2 - 2058\,X - 441$$

Donc la suite de polynômes est périodique à partir du 2ème terme, de période 24. Les x_n ne sont autres que les racines positives des P_n, donc le développement en fraction continue de x est périodique à partir du 2ème terme, de période 24.
D'ailleurs :

```
> x:=2/3+5/7*sqrt(13):T:=x:
  for i to 25 do x:=factor(1/(x-floor(x)));T:=T,x od:
  T;
```

$\frac{2}{3}+\frac{5\sqrt{13}}{7}, \frac{1029}{524}+\frac{315\sqrt{13}}{524}, \frac{1067}{289}+\frac{315\sqrt{13}}{289}, \frac{956}{1301}+\frac{315\sqrt{13}}{1301}, \frac{23}{60}+\frac{7\sqrt{13}}{20}, \frac{555}{1091}+\frac{315\sqrt{13}}{1091}, \frac{536}{919}+\frac{315\sqrt{13}}{919},$
$\frac{383}{1244}+\frac{315\sqrt{13}}{1244}, \frac{41}{21}+\frac{5\sqrt{13}}{7}, \frac{903}{1076}+\frac{315\sqrt{13}}{1076}, \frac{173}{1171}+\frac{315\sqrt{13}}{1171}, \frac{998}{251}+\frac{315\sqrt{13}}{251}, \frac{202}{215}+\frac{63\sqrt{13}}{215}, \frac{5}{92}+\frac{315\sqrt{13}}{1196},$
$\frac{377}{3}+35\sqrt{13}, \frac{1128}{1949}+\frac{315\sqrt{13}}{1949}, \frac{821}{316}+\frac{315\sqrt{13}}{316}, \frac{43}{17}+\frac{63\sqrt{13}}{85}, \frac{50}{21}+\frac{5\sqrt{13}}{7}, \frac{714}{1769}+\frac{315\sqrt{13}}{1769}, \frac{211}{20}+\frac{63\sqrt{13}}{20},$
$\frac{1045}{1979}+\frac{315\sqrt{13}}{1979}, \frac{934}{211}+\frac{315\sqrt{13}}{211}, \frac{193}{340}+\frac{63\sqrt{13}}{340}, \frac{5}{3}+\frac{5\sqrt{13}}{7}, \frac{1029}{524}+\frac{315\sqrt{13}}{524}$

et on voit bien que le dernier terme est égal au deuxième.

```
>
```